
D05 – Integral Equations

D05AAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

D05AAF solves a linear, non-singular Fredholm equation of the second kind with a split kernel.

2 Specification

SUBROUTINE D05AAF(LAMBDA, A, B, K1, K2, G, F, C, N, IND, W1, W2,
1 WD, NMAX, MN, IFAIL)
INTEGER N, IND, NMAX, MN, IFAIL
real LAMBDA, A, B, K1, K2, G, F(N), C(N),
1 W1(NMAX,MN), W2(MN,4), WD(MN)
EXTERNAL K1, K2, G

3 Description

D05AAF solves an integral equation of the form

f(x)− λ

∫ b

a

k(x, s)f(s) ds = g(x)

for a ≤ x ≤ b, when the kernel k is defined in two parts: k = k1 for a ≤ s ≤ x and k = k2 for x < s ≤ b.
The method used is that of El-gendi [2] for which, it is important to note, each of the functions k1 and
k2 must be defined, smooth and non-singular, for all x and s in the interval [a, b].

An approximation to the solution f(x) is found in the form of an n term Chebyshev-series
n∑

i=1

′ciTi(x),

where ′ indicates that the first term is halved in the sum. The coefficients ci, for i = 1, 2, .., n, of this
series are determined directly from approximate values fi, for i = 1, 2, . . . , n, of the function f(x) at the
first n of a set of m+ 1 Chebyshev points:

xi =
1
2
(a+ b+ (b − a) cos[(i − 1)π/m]), , i = 1, 2, . . . , m+ 1.

The values fi are obtained by solving simultaneous linear algebraic equations formed by applying a
quadrature formula (equivalent to the scheme of Clenshaw and Curtis [1]) to the integral equation at the
above points.

In general m = n − 1. However, if the kernel k is centro-symmetric in the interval [a, b ], i.e., if
k(x, s) = k(a+b−x, a+b−s), then the routine is designed to take advantage of this fact in the formation
and solution of the algebraic equations. In this case, symmetry in the function g(x) implies symmetry
in the function f(x). In particular, if g(x) is even about the mid-point of the range of integration, then
so also is f(x), which may be approximated by an even Chebyshev-series with m = 2n − 1. Similarly, if
g(x) is odd about the mid-point then f(x) may be approximated by an odd series with m = 2n.

4 References

[1] Clenshaw CW and Curtis A R (1960) A method for numerical integration on an automatic computer
Numer. Math. 2 197–205

[2] El–Gendi S E (1969) Chebyshev solution of differential, integral and integro-differential equations
Comput. J. 12 282–287

5 Parameters

1: LAMBDA — real Input

On entry: the value of the parameter λ of the integral equation.

2: A — real Input

On entry: the lower limit of integration, a.

3: B — real Input

On entry: the upper limit of integration, b.

Constraint: B > A.

4: K1 — real FUNCTION, supplied by the user. External Procedure

K1 must evaluate the kernel k(x, s) = k1(x, s) of the integral equation for a ≤ s ≤ x.

Its specification is:

real FUNCTION K1(X, S)
real X, S

1: X — real Input
2: S — real Input

On entry: the values of x and s at which k1(x, s) is to be evaluated.

K1 must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters
denoted as Input must not be changed by this procedure.

5: K2 — real FUNCTION, supplied by the user. External Procedure

K2 must evaluate the kernel k(x, s) = k2(x, s) of the integral equation for x < s ≤ b.

Its specification is:

real FUNCTION K2(X, S)
real X, S

1: X — real Input
2: S — real Input

On entry: the values of x and s at which k2(x, s) is to be evaluated.

Note that the functions k1 and k2 must be defined, smooth and non-singular for all x and s in the
interval [a, b].

K2 must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters
denoted as Input must not be changed by this procedure.

6: G — real FUNCTION, supplied by the user. External Procedure

G must evaluate the function g(x) for a ≤ x ≤ b.

Its specification is:

real FUNCTION G(X)
real X

1: X — real Input
On entry: the values of x at which g(x) is to be evaluated.

G must be declared as EXTERNAL in the (sub)program from which D05AAF is called. Parameters
denoted as Input must not be changed by this procedure.

7: F(N) — real array Output

On exit: the approximate values fi, for i = 1,2,...,N of f(x) evaluated at the first N of M + 1
Chebyshev points xi, (see Section 3).

If IND is 0 or 3, M = N− 1; if IND is 1, M = 2×N and if IND is 2, M = 2×N− 1.

8: C(N) — real array Output

On exit: the coefficients ci, for i = 1,2,...,N of the Chebyshev-series approximation to f(x).

If IND is 1 this series contains polynomials of odd order only and if IND is 2 the series contains
even order polynomials only.

9: N — INTEGER Input

On entry: the number of terms in the Chebyshev-series required to approximate f(x).

10: IND — INTEGER Input

On entry: IND must be set to 0,1,2 or 3.

IND = 0

k(x, s) is not centro-symmetric (or no account is to be taken of centro-symmetry).
IND = 1

k(x, s) is centro-symmetric and g(x) is odd.
IND = 2

k(x, s) is centro-symmetric and g(x) is even.
IND = 3

k(x, s) is centro-symmetric but g(x) is neither odd nor even.

11: W1(NMAX,MN) — real array Workspace
12: W2(MN,4) — real array Workspace
13: WD(MN) — real array Workspace

14: NMAX — INTEGER Input

On entry: the first dimension of the array W1 as declared in the (sub)program from which D05AAF
is called.

Constraint: NMAX ≥ N.

15: MN — INTEGER Input

On entry: the first dimension of the array W2 as declared in the (sub)program from which D05AAF
is called.

Constraint: MN ≥ 2×N+ 2.

16: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

A ≥ B.

IFAIL = 2

A failure has occurred (in F04AAF unless N = 1) due to proximity to an eigenvalue. In general, if
LAMBDA is near an eigenvalue of the integral equation, the corresponding matrix will be nearly
singular.

7 Accuracy

No explicit error estimate is provided by the routine but it is usually possible to obtain a good indication
of the accuracy of the solution either

(i) by examining the size of the later Chebyshev coefficients ci, or
(ii) by comparing the coefficients ci or the function values fi for two or more values of N.

8 Further Comments

The time taken by the routine increases with N.

This routine may be used to solve an equation with a continuous kernel by calling the same FUNCTION
for K2 as for K1.

This routine may also be used to solve a Volterra equation by defining K2 (or K1) to be identically zero.

9 Example

The example program solves the equation

f(x)−
∫ 1

0

k(x, s)f(s) ds = (1 − 1
π2
) sin(πx)

where
k(x, s) =

{
s(1− x) for 0 ≤ s < x,
x(1 − s) for x ≤ s ≤ 1.

Five terms of the Chebyshev-series are sought, taking advantage of the centro-symmetry of the k(x, s)
and even nature of g(x) about the mid-point of the range [0, 1].

The approximate solution at the point x = 0.1 is calculated by calling C06DBF.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* D05AAF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER N, NMAX, MN
PARAMETER (N=5,NMAX=N,MN=2*N+2)
INTEGER NOUT
PARAMETER (NOUT=6)

* .. Scalars in Common ..
real R

* .. Local Scalars ..
real A, ANS, B, LAMBDA, X
INTEGER I, IFAIL, IND, IS

* .. Local Arrays ..
real C(NMAX), F(NMAX), W1(NMAX,MN), W2(MN,4), WD(MN)

* .. External Functions ..
real C06DBF, G, K1, K2, X01AAF
EXTERNAL C06DBF, G, K1, K2, X01AAF

* .. External Subroutines ..
EXTERNAL D05AAF

* .. Common blocks ..
COMMON R

* .. Executable Statements ..
WRITE (NOUT,*) ’D05AAF Example Program Results’
WRITE (NOUT,*)
R = X01AAF(0.0e0)
LAMBDA = 1.0e0
A = 0.0e0
B = 1.0e0
IND = 2
IFAIL = 0
WRITE (NOUT,*)

+’Kernel is centro-symmetric and G is even so the solution is even’
WRITE (NOUT,*)

*
CALL D05AAF(LAMBDA,A,B,K1,K2,G,F,C,N,IND,W1,W2,WD,NMAX,MN,IFAIL)

*
WRITE (NOUT,*) ’Chebyshev coefficients’
WRITE (NOUT,*)
WRITE (NOUT,99998) (C(I),I=1,N)
WRITE (NOUT,*)
X = 0.1e0

* Note that X has to be transformed to range [-1,1]
IS = 1
IF (IND.EQ.1) THEN

IS = 3
ELSE

IF (IND.EQ.2) IS = 2
END IF
ANS = C06DBF(2.0e0/(B-A)*(X-0.5e0*(B+A)),C,N,IS)
WRITE (NOUT,99999) ’X=’, X, ’ ANS=’, ANS
STOP

*
99999 FORMAT (1X,A,F5.2,A,1F10.4)
99998 FORMAT (1X,5e14.4)

END
*

real FUNCTION K1(X,S)
* .. Scalar Arguments ..

real S, X
* .. Executable Statements ..

K1 = S*(1.0e0-X)
RETURN
END

*
real FUNCTION K2(X,S)

* .. Scalar Arguments ..
real S, X

* .. Executable Statements ..
K2 = X*(1.0e0-S)
RETURN
END

*
real FUNCTION G(X)

* .. Scalar Arguments ..
real X

* .. Scalars in Common ..
real R

* .. Intrinsic Functions ..
INTRINSIC SIN

* .. Common blocks ..
COMMON R

* .. Executable Statements ..
G = SIN(R*X)*(1.0e0-1.0e0/(R*R))
RETURN
END

9.2 Program Data

None.

9.3 Program Results

D05AAF Example Program Results

Kernel is centro-symmetric and G is even so the solution is even

Chebyshev coefficients

0.9440E+00 -0.4994E+00 0.2799E-01 -0.5967E-03 0.6658E-05

X= 0.10 ANS= 0.3090

